Plan payments and savings in Excel 2010
Quick Reference Card

How to enter functions into formulas
On the Formulas tab, click Insert Function. In the Search for a function box, type what
you are looking for.

For example, “Monthly payments,” and then click GO. Select a function in the list and read
its description. If it’s the function you are looking for, click OK to open the Function
Arguments dialog box. Then enter the arguments and click OK.
If you already know the name of the function you want, but want help with the arguments,
type the name of the function in the Search for a function box, and then click GO. Select the
function in the list, and then click OK to open the Function Arguments dialog box.
OR
If you know the name of the function you want to use, you can type an equal (=) sign in any
empty cell in your spreadsheet. Then begin to type the function name. Formula
AutoComplete will offer you a list of functions.

Double-click the one you want to let Excel enter the function name and the formula’s
beginning parenthesis into the spreadsheet. After the function is entered in your spreadsheet,
a ScreenTip will show you the arguments you need to enter.

Using the PMT function
The PMT function calculates the payment for a loan based on constant payments and a
constant interest rate. Here are some examples of how to use the PMT function from the
course.
Pay off a credit card balance
Assume that the balance due is $5,400 at a 17% annual interest rate. Nothing else will be
purchased on the card while the debt is being paid off.
=PMT(17%/12,2*12,5400) The result is a monthly payment of $266.99 to pay the debt off in
two years.




The Rate argument is the interest rate per period for the loan. For example, in this
formula the 17% annual interest rate is divided by 12, the number of months in a year.
The NPER argument of 2*12 is the total number of payment periods for the loan.
The PV or present value argument is 5400.

Figure out monthly mortgage payments
Imagine a $180,000 home at 5% interest, with a 30-year mortgage.
=PMT(5%/12,30*12,180000) The result is a monthly payment (not including insurance and
taxes) of $966.28.




The Rate argument is 5% divided by the 12 months in a year.
The NPER argument is 30*12 for a 30 year mortgage with 12 monthly payments
made each year.
The PV argument is 180000 (the present value of the loan).

Figure how much to save for a vacation
You’d like to save for a vacation three years from now that will cost $8,500. The annual
interest rate for saving is 1.5%.
=PMT(1.5%/12,3*12,0,8500) To save $8,500 in three years would require a savings of
$230.99 each month for three years.





The Rate argument is 1.5% divided by 12, the number of months in a year.
The NPER argument is 3*12 for twelve monthly payments over three years.
The PV (present value) is 0 because the account is starting from zero.
The FV (future value) that you want to save is $8,500.

Using the PV function

The PV function returns the present value of an investment. The present value is the total
amount that a series of future payments is worth now. Here are some examples of how to use
the PV function from the course.
Figure out how much house you can afford to buy
Say that the interest rate is 5%, and that you can afford an $800 monthly mortgage payment,
not including taxes and insurance.
=PV(5%/12,30*12,-800) The formula result is that an affordable house would cost
$149,025.29.




The Rate argument is 5% divided by 12 months of the year.
The NPER argument is 30*12 (or twelve monthly payments for 30 years).
The PMT is -800 (you would pay out $800 a month).

See how a starting deposit affects savings
Imagine that you are saving for an $8,500 vacation over three years, and wonder how much
you would need to deposit in your account to keep monthly savings at $175.00 per month.
The PV function will calculate how much of a starting deposit will yield a future value.
=PV(1.5%/12,3*12,-175,8500) An initial deposit of $1,969.62 would be required in order to
be able to pay $175.00 per month and end up with $8500 in three years.





The rate argument is 1.5%/12.
The NPER argument is 3*12 (or twelve monthly payments for three years).
The PMT is -175 (you would pay $175 per month)
The FV (future value) is 8500

Figure out a down payment
Say that you’d like to buy a $19,000 car at a 2.9% interest rate over three years. You want to
keep the monthly payments at $350 a month, so you need to figure out your down payment.
In this formula the result of the PV function is the loan amount, which is then subtracted from
the purchase price to get the down payment.
=19000-PV(2.9%/12, 3*12,-350) The down payment required would be $6,946.48





The $19,000 purchase price is listed first in the formula. The result of the PV function
will be subtracted from the purchase price.
The rate argument is 2.9% divided by 12.
The NPER argument is 3*12 (or twelve monthly payments over three years).
The PMT is -350 (you would pay $350 per month).

Using the NPER function
The NPER function returns the number of periods for an investment based on periodic,
constant payments and a constant interest rate. Here is an example of using the function from
the course.

Figure out how long it will take to pay off a personal loan
Imagine that you have a $2,500 personal loan, and have agreed to pay $150 a month at 3%
annual interest.
=NPER(3%/12,-150,2500) It would take 17 months and some days to pay off the loan.




The rate argument is 3%/12 monthly payments per year.
The PMT argument is -150
The PV (present value) argument is 2500

Tip
Because the payment period doesn’t come out to an even amount, you’d want to round up to
18 months. You could do that using the PMT function.
=PMT(3%/12,18,2500) Results in 18 payments of $142.21 per month
In this formula the rate argument is 3%/12, the NPER (number of payments) argument is 18,
and the PV or present value argument is $2,500.

Using the FV function
The FV function returns the future value of an investment based on periodic, constant
payments and a constant interest rate. Here is an example of the function from the course.
See how much your savings will add up to over time
Starting with $500 in your account, how much will you have in 10 months if you deposit
$200 a month at 1.5% interest?
=FV(1.5%/12,10,-200,-500) In 10 months you would have $2,517.57 in savings.





The rate argument is 1.5%/12.
The NPER argument is 10 (months)
The PMT argument is -200
The PV (present value) argument is -500

